
1. Introduccion

Prolog deriva su nombre de Programming in logic, es decir, la idea surgida en los afios 70 de usar
a la l6gica como lenguaje de programacion. Prolog es un lenguaje de programacion centrado alrededor
de un conjunto pequefio de mecanismos basicos que incluyen: concordancia (matching) de patrones,
estructuracion de datos basada en drboles y backtracking automaético.

Prolog es un lenguaje pensado para la computaciéon simbdlica, no numérica. Se adapta particu-
larmente bien a problemas que involucran objetos y relaciones entre objetos. Estas caracteristicas han
convertido a Prolog en un lenguaje poderoso para aplicaciones de Inteligencia Artificial y programacién
no numérica en general.

2. Definicion de relaciones mediante hechos

La forma maés sencilla de especificar una relaciéon en Prolog es mediante un conjunto de hechos.
Cada hecho consta del nombre de la relacién (que comienza con una letra mindscula), seguido por
una lista de argumentos separados por coma y encerrados entre paréntesis. Cada hecho finaliza con un
punto. Ejemplo: el hecho de que Tom es progenitor de Bob se puede escribir en Prolog como:

progenitor (tom, bob) .

En este caso hemos elegido progenitor como nombre de la relacion (binaria) y a tom y bob
como sus argumentos.



Figura 1: El drbol familiar.

Para definir una relacion familiar completa como la que se muestra en la Figura 1, se podria usar el
siguiente programa Prolog:

progenitor (pam, bob
progenitor (tom, bob
progenitor (tom, liz
progenitor (bob, ann
progenitor (bob, pat
progenitor (pat, jim

)
)
) .
) .
)
)

Una vez que este programa ha sido comunicado al sistema Prolog, se pueden realizar distintas
consultas referidas a la relacion progenitor. Por ejemplo, para preguntar: ;Bob es progenitor de
Pat?, 1a consulta puede ser realizada al sistema Prolog tipeando en la terminal:

?— progenitor (bob, pat) .
A lo que Prolog respondera:

yes

Otra consulta puede ser:

?— progenitor (liz,pat).
A lo que Prolog respondera:

no
o bien

?— progenitor (tom, ben) .
no

La pregunta ;Quién es el progenitor de Liz? puede ser realizada tipeando:



?— progenitor (X,1liz).

En este caso, Prolog no sélo responde yes o no, sino que responde cual es el valor de X que hace la
sentencia previa verdadera. La respuesta es:

X=tom

La pregunta ;Quiénes son los hijos o hijas de Bob? puede ser comunicada a Prolog como:
?— progenitor (bob, X) .

En este caso hay mds de una respuesta. La primera respuesta de Prolog es:

X=ann

T3]

Si requerimos otra solucion, tipeamos un punto y coma (‘“;”’) y Prolog responde:
X=pat

Si solicitamos otra respuesta (tipeando un ;) la respuesta serd no. También es posible preguntar ;Quién
es progenitor y de quienes?. Otra forma de plantear la pregunta es diciendo: “Encontrar los X e Y tal
que X es progenitor de Y".

?— progenitor (X,Y).
X=pam
Y=bob;

X=tom
Y=bob;

X=tom
Y=1iz;

La muestra de soluciones puede ser detenida ingresando un return en lugar de un punto y coma. Para
preguntar ;Quién es abuelo o abuela de Jim?, debemos descomponer la consulta en dos pasos:

(Quién es un progenitor de Jim?. Asumamos que es algin Y.

(Quién es un progenitor de Y?. Asumamos que es algin X.

?— progenitor (Y, jim),progenitor (X,Y) .
X=bob
Y=pat

La “) se lee como “y” (“A” 16gico). Alterando el orden de las componentes de la consulta, el significado
16gico es el mismo y producird el mismo resultado.

Ejercicio: Plantear las consultas que permitan saber: a) ;Quiénes son lon nietos de Tom? y b) ;jAnn
y Pat tienen algun progenitor en comun?



3. Definicion de relaciones mediante reglas

Comencemos agregando informacién sobre el sexo de las personas del ejemplo anterior:

mujer (pam) .
hombre (tom) .
hombre (bob) .
mujer (liz).
mujer (pat) .
mujer (ann) .
hombre (jim) .

Las relaciones mujer y hombre son relaciones unarias, mientras que la relacion progenitor era
binaria, ya que relacionaba pares de objetos.

Si quiciéramos establecer la relaciéon descendiente, vemos que ésta es la relacion inversa de
progenitor. Una alternativa para la definicion de esta relacion consistiria en ingresar una lista de
hechos:

descendiente (bob, pam) .
descendiente (bob, tom) .

que es igual a la relacion progenitor pero con sus argumentos invertidos.
Sin embargo, si la relacién progenitor ya estd definida, y sabiendo que descendiente es la
relacidn inversa, esta relacion puede ser definida en base a la siguiente sentencia logica:

Paratodo X e Y
Y es descendiente de X si X es progenitor de Y.

Esta sentencia ldgica tiene un formato similar al utilizado por Prolog. La cldusula Prolog que tiene
el mismo significado es:

descendiente (Y, X) :- progenitor(X,Y).

Este tipo de cldusulas se denominan reglas y también se pueden leer como:

Paratodo X e Y
Si X es progenitor de Y entonces Y es descendiente de X .

9

Las reglas especifican cosas que son verdaderas si alguna condicion es satisfecha. El simbolo “: -
(que se lee “if” o “si”’) delimita dos partes dentro de una regla, una izquierda y otra derecha. Asi, en
una regla podemos diferenciar:

= una parte de condicion (la parte derecha de la regla).
= una parte de conclusion (la parte izquierda de la regla).

La parte de la conclusién también es llamada la cabeza de una cldusula y la parte de condicidn el cuerpo
de la clausula. En el ejemplo de la regla anterior descendiente (Y, X) constituye la cabeza de la



regla y progenitor (X, Y) es el cuerpo de la misma. Esta regla puede ser interpretada como una
sentencia que expresa que si la condicion progenitor (X, Y) es verdadera, una consecuencia logica
de ésto es descendiente (Y, X).

Para comprender la forma en que Prolog utiliza las reglas, veamos que sucede cuando realizamos la
siguiente consulta:

?— descendiente (liz,tom) .

Ningtn hecho habla sobre la relacién descendiente por lo que Prolog sélo podra dar una respuesta
en base a la regla definida previamente. Esta regla es general, en el sentido que puede ser aplicada a
cualquier objeto X e Y y por consiguiente puede ser aplicada a objetos particulares como por ejemplo
1izy tom. Paraaplicarlareglaa 11z y tom, Y debe ser substituido con 11z, y X con tom. Decimos
en este caso, que las variables X e Y se han instanciado a:

X=tom , Y=liz
Luego de la instanciacion hemos obtenido un caso especial de la regla general:
descendiente(liz,tom) :- progenitor (tom,liz).

La parte de la condicién es ahora progenitor (tom, 1iz). Prolog intenta ahora descubrir si la parte
de la condicion es verdadera. Asi, el objetivo inicial descendiente (1iz, tom) ha sido reemplaza-
do con el subobjetivo progenitor (tom, 1iz). Este (nuevo) objetivo es trivial ya que se encuentra
como un hecho en nuestro programa. Esto significa que la parte de la conclusion también es verdadera
y por lo tanto Prolog respondera la consulta con yes.

Ejercicio: Determinar cuales son las relaciones familiares que determinan las siguientes reglas:

ma (X,Y) :— progenitor(X,Y), mujer (X).
ab(X,Z2) :- progenitor(X,Y) , progenitor (Y,Z2).

3.1. Reglas recursivas

Supongamos que deseamos agregar a nuestro programa de familia la relacidn predecesor. Si defini-
mos esta relacion en términos de la relacidn progenitor, vemos que esta relacion tiene un caso elemental
que se da cuando una persona X es predecesora de una persona Z dado que X es progenitor de Z. En
Prolog:

predecesor (X,Z) :— progenitor(X,Z72).

Esta regla sélo sirve para el caso en que X sea predecesor directo de Z. Pero X puede ser un predece-
sor indirecto de Z, es decir, puede existir una cadena de progenitores que conectan X con Z. Un intento
para resolver ésto, seria con el siguiente programa:

predecesor (X,2) :-—
progenitor (X, 7).

predecesor (X,Z) :-—
progenitor (X,Y),
progenitor (Y, 7).



predecesor (X,72) :-—
progenitor (X,Y1),
progenitor (Y1l,Y2),
progenitor (Y2, 7).

predecesor (X, Z) :-—
progenitor (X, Y1),
progenitor(Y1l,Y2),
progenitor (Y2, Y3)
progenitor (Y3, 7).

’

El programa anterior tiene dos problemas. Por un lado es muy tedioso para escribir. Por otra parte,
sOlo detectard los predecesores hasta una determinada profundidad. Una solucién a este problema con-
siste en definir la regla en términos de si mismo, es decir, en forma recursiva. De esta forma, la relacion
predecesor se puede definir:

predecesor (X,72) :-—
progenitor (X, 7).

predecesor (X,72) :-—
progenitor (X,Y),
predecesor (Y, 7).

Como vemos, la primera regla servird para manejar los casos en que la relacion predecesor es
directa. Usando la terminologia de recursion, esta regla actuard como punto conocido del dominio de
la definicién recursiva. Por otra parte, la segunda regla manejard los casos en que existe una relacion
predecesor indirecta, y servird para distintos niveles de profundidad entre dos personas.

Ejercicio: Especificar una consulta que determine las personas de las cuales Pam es predecesor (o
sea, los sucesores de Pam), y describir cuales serdn las respuestas de Prolog.

4. Variables anonimas

Hasta ahora hemos visto que en las cldusulas pueden aparecer &tomos simples como tomy jim
y variables como X e Y. Los nombres de las variables pueden tener en realidad un nimero arbitrario
de caracteres, pero deben comenzar con una letra maytscula. Existen casos, donde el nombre de una
variable puede ser irrelevante. Por ejemplo, para definir una regla que determine si una persona tiene
un hijo, se podria escribir:

tiene_un_hijo(X) :- progenitor(X,Y).

Observar que la propiedad de que alguien tenga un hijo, es independiente del nombre del hijo. En estos
casos, no tiene sentido inventar un nombre de variable para algo que no nos interesa. Una alternativa
consiste en utilizar un tipo de variable especial denominada variable andnima que se denota con el
simbolo “_” (underscore, o “guion bajo”). Usando una variable anénima, la regla anterior se puede
escribir:

tiene_un_hijo(X) :- progenitor(X,_).



Una regla util para determinar si una variable deberia ser reemplazada con una variable anénima, es
observar si dicha variable aparece en la clausula una tinica vez. Las variables an6nimas también pueden
aparecer en las consultas y los valores que toman no son mostrados por el sistema Prolog. Por ejemplo,
para determinar todas las personas que son progenitoras de alguien, puedo escribir:

?— progenitor (X,_) .
X=pam;
X=tom;
X=tom;
X=bob;

5. Objetos de datos

En la Figura 2 se muestra una clasificacion de los objetos de datos en Prolog:

objetos de datos

/

objetos simples estructuras

N

constantes variables

N\

atomos numeros
Figura 2: Objetos de datos en Prolog

Atomos y niimeros
Los dtomos se pueden construir de tres maneras:

6

1. Cadenas de letras, digitos y el caracter underscore (
Ejemplo: ana, x25, juan_perez.

), comenzando con una letra minuscula.

2. Cadenas de caracteres especiales. Ejemplo: ==>, ><:>, . ;. >.
3. Cadenas de caracteres encerrada entre comillas simples. Ejemplo: *Ana’.

Los numeros utilizados en Prolog incluyen a los enteros (p.ej. 1, 1313, -97) y los reales (p.ej. 3.14,
-0.035).



5.1. Variables

Las variables son cadenas de letras, digitos y caracteres underscore. Las variables comienzan con
una letra mayuscula o un caracter underscore. Ejemplo: X, Result, Pepe2,_x23,_23. Observar
que las variables andnimas son variables especiales que s6lo contienen el caracter *“_”

El alcance lexicogrdfico de una variable estd delimitado por la cldusula que la contiene (un hecho,
una regla o una consulta). Esto significa que si el mismo nombre de variable aparece en dos reglas
distintas, representa en realidad dos variables distintas. Lo mismo ocurre si el mismo nombre de variable
ocurre en una consulta y en una regla o hecho.

5.2. Numeros y Aritmética en Prolog

A partir de la disponibilidad de variables y ntimeros, es posible realizar operaciones aritméticas en
Prolog que tienen ciertas particularidades. Supongamos la siguiente consulta en Prolog:

- X =1+ 2.

X =1+ 2
El resultado no es X = 3 como uno esperaria. El problema es que el operador “=" trata de igualar dos
términos pero no fuerza la evaluacion de operaciones. Para lograr dicho efecto, se utiliza el operador

G‘i S”:

- X i1s 1 + 2.
X =3

(1321

El operador “is” también permite evaluar otras operaciones como la resta (
division(*/”), potencia(“**””) y médulo(“mod”). Ejemplo:

), multiplicacion(‘“*”),

?—- X is 5/2, Y is 2 ** 3, Z 1is 5 mod 2.
X = .5

Y:
Z

= 0 N

Existen operadores que permiten comparar valores numéricos y que automaticamente fuerzan la eva-
luacién de los términos. Algunas operaciones relacionales vdlidas son la de mayor (“>”"), menor (<),
mayor o igual (“>=") y menor o igual (“=<"). Ejemplo:

?2— 3 x 4 > 10.
yes

El operador “\="sirve para controlar si dos términos son distintos.



5.3. Estructuras

Los objetos estructurados (o simplemente estructuras) son objetos que tienen varias componentes.
Las componentes de una estructura pueden ser objetos simples (dtomos, nimeros o variables) o bien
otras estructuras. Una estructura se representa con un functor seguido por las componentes (o argumen-
tos) separados por comas y encerrados entre paréntesis. Ejemplo:

fecha (8, mayo, 2003)

Observar la similitud entre la sintaxis de las estructuras y la utilizada para representar las relaciones.
Sin embargo, es importante diferenciarlos, ya que las estructuras al igual que otro tipo de objetos como
variables, &tomos y nimeros siempre las utilizaremos en los argumentos de las relaciones.

Las estructuras se pueden representar como drboles. La raiz del arbol es el functor y los descendien-
tes de la raiz son sus componentes. Asi, la estructura fecha (8, mayo, 2003) puede ser representada
como el arbol que se muestra en la Figura 3.

fecha

8 mayo 2003

Figura 3: La estructura fecha (8, mayo, 2003) representada como drbol.

Para observar como pueden estar anidadas las estructuras, veamos un ejemplo donde se utilizan
estructuras para representar objetos geométricos simples. Un punto en un espacio de dos dimensiones
estd definido por dos coordenadas; una linea estd definida por dos puntos; y un tridngulo se puede definir
con tres puntos. Para representar este tipo de objetos se podrian utilizar estructuras como las mostradas
a continuacion:

punto(1l,1) punto (2, 3)
linea(punto(1l,1), punto(2,3))

triangulo (punto(4,2), punto(6,4), punto(7,1))

Ejercicio: Dibuje los objetos representados por estas estructuras, y dé las representaciones de arbol
correspondiente.

En el caso de estructuras anidadas, el functor que aparece en la raiz del 4rbol se denomina functor
principal.



6. Matching (coincidencia) de términos

Todos los tipos de objetos que vimos previamente se denominan términos. La operaciéon mas im-
portante que se puede realizar sobre términos se denomina matching. En forma informal se puede decir
que esta operacion determina si dos términos coinciden o en otras palabras hacen juego. Dados dos
términos, decimos que éstos hacen matching si:

1. los términos son idénticos, o

2. las variables en ambos términos pueden ser instanciadas a objetos de tal manera que después de
la substitucion de las variables por estos objetos los términos se tornan idénticos.

Los términos fecha (D, M, 2003) y fecha (D1, mayo, Y1) son un ejemplo de dos estructuras
que hacen matching. Una instanciacion que hace estos términos idénticos es:

1. D es instanciada con D1.
2. M es instanciada con mayo.
3. Y1 es instanciada con 2003.

De acuerdo al formato utilizado por Prolog para mostrar los resultados, esta instanciacién seria:

D = D1
M = mayo
Y1l = 2003

El proceso de matching toma como entrada dos términos y controla si los mismos “hacen juego”.
Cuando los términos no coinciden se dice que el proceso de matching fallo. Si en cambio los términos
coinciden entonces se dice que el proceso fuvo éxito y se instancian las variables en ambos términos a
valores que hacen los términos idénticos.

El proceso de matching se realiza siempre que se utiliza el operador

“__9

en una consulta:

?— fecha(D,M,2003) = fecha(Dl,mayo,Y1l).
D = D1

M = mayo

Yl = 2003

La instanciacion de variables que muestra Prolog no es la unica posible. Otras instanciaciones véli-
das podrian ser:

D =1
D1 =1
M = mayo
Yl = 2003

D = pirulo
D1 = pirulo
M = mayo
Yl = 2003



Prolog siempre muestra la instanciacion mds general. Esta instanciacion es aquella que compromete las
variables en la menor medida posible.

Al utilizarse la instanciacion mds general se deja el mayor grado de libertad posible para realizar
otras instanciaciones en el caso de que se requieran otras operaciones de matching.

Ejemplo:

?— fecha(D,M,2003) = fecha(Dl,mayo,Y1l),
fecha (D,M, 2003) fecha (15,M,Y).

Para satisfacer el primer objetivo, Prolog internamente instancia las variables de la siguiente manera:

D = D1
M = mayo
Yl = 2003

Sin embargo después de satisfacer el segundo objetivo la instanciacion se vuelve mas especifica y
el resultado de la consulta es:

D =15

D1 = 15
M = mayo
Yl = 2003
Y = 2003

Esto muestra que durante la ejecucion de objetivos consecutivos, las variables usualmente se van ins-
tanciando con valores que son incrementalmente mds especificos.

6.1. Reglas de Matching

Las reglas generales para determinar si dos términos S'y 7" hacen matching son las siguientes:
1. Si S'y T son constantes, S'y 7" hacen matching sélo si son el mismo objeto.

2. Si S es una variable y 7" es cualquier cosa, los términos hacen matching y S es instanciada a 7.
A su vez, si T es una variable entonces 7' es instanciada a S.

3. Si Sy T son estructuras, éstas hacen matching si:

a) Sy T tienen el mismo functor principal, y

b) Las componentes correspondientes hacen matching.
La instanciacion resultante es determinada por el matching de cada una de las componentes.
Ejemplo:

?— triangulo (punto(l,1),A,punto(2,3)) =
triangulo (X,punto(4,Y),punto(2,2)) .

X = punto(l,1)

A punto (4,Y)

Z =3



Hasta el momento sélo hemos observado la forma en que se realiza el proceso de matching cuando
se utiliza el operador “=". Sin embargo, este proceso se realiza repetidamente durante el funcionamiento
del intérprete Prolog. En particular, el matching de términos se realiza cada vez que un objetivo debe
ser probado y se busca un hecho o la cabeza de una regla que coinicidan con este objetivo.

Ejemplo: Supongamos que se tiene el siguiente programa Prolog, que permite determinar si una
linea es horizontal o vertical:

vertical (1linea (punto(X,_),punto(X,_))) .
horizontal (linea (punto(_,Y),punto(_,Y))).

“Vertical” es una propiedad de las lineas y es representada en este caso como una relacion unaria.
Vemos en este caso que una linea es vertical si las coordenadas x de sus puntos extremos son iguales.
Consideraciones similares se pueden realizar con el predicado “horizontal”. Ahora es posible realizar
la siguiente conversacion con este programa:

?— vertical (linea (punto(1l,1),punto(1,2))).
yes

?— vertical (linea (punto(1l,1),punto(2,Y))).
no

?— horizontal (linea (punto(1l,1),punto(2,Y))).
Yy =1

Una pregunta que podemos realizar es: ;hay alguna linea vertical que comienza en el punto (2,3)?:

?— vertical (linea (punto(2,3),P)).
P = punto(2,_)

La respuesta en este caso significa que si, y dice ademds que serd cualquier linea que finaliza en algin
punto (2,_) (la linea vertical x = 2). Otra consulta interesante seria preguntar si existe alguna linea que
es simultdneamente vertical y horizontal:

?— vertical (L) , horizontal(L).
L = linea(punto(X,Y),punto(X,Y))



7. Listas

Las listas constituyen las estructuras mas simples y ttiles que se pueden utilizar en Prolog. Una lista
es una secuencia de cualquier nimero arbitrario de items. Una lista con los items ana, tenis, tom, sky
se puede escribir en Prolog como:

[ana, tenis, tom, sky]

Para representar listas se deben considerar dos casos: la lista es vacia o no vacia. Para el primer
caso, se utiliza un dtomo especial de Prolog: []. Cuando la lista no es vacia, se puede decir que ésta
consiste de dos cosas:

1. el primer item, llamado la cabeza de la lista;
2. el resto de la lista, llamado la cola.

En la lista de ejemplo, ana es la cabezay [tenis, tom, sky] es la cola. En general, la cabeza
puede ser cualquier objeto Prolog pero la cola siempre sera una lista.

Si bien la notacion utilizada con la lista de ejemplo no respeta el formato de las estructuras, es
importante notar que las listas son en realidad estructuras. Una lista puede ser representada combinando
la cabeza y la cola con un functor especial (*.”):

. (Cabeza, Cola)
Asi, la lista de ejemplo podria haber sido representada con esta notacion de la siguiente manera:
. (ana, . (tenis, . (tom, . (sky, [1))))

La notacién [Iteml, Item2,....] es la usada normalmente en los programas Prolog. No
obstante esto, es importante observar que sé6lo se trata de una mejora cosmética en la sintaxis. Interna-
mente, las listas son estructuras de dos componentes que se representan mediante drboles binarios.

Prolog provee otra extension notacional que permite diferenciar la cabeza y la cola de una lista,
utilizando la barra vertical (“I”’):

[Cabeza|Cola]

Con esta notacion es posible referenciar la cola completa de una lista como un objeto simple. Ejemplo:

- [1,2,3] = [C|T].
=1
T = 1[2,3]
?= [a,b,c] = [allbllc]]].
yes
?— [ana,pepe,tom, lucas] = [A]| [pepelB]l].
A = ana
B = [tom, lucas]



La notacién con barra vertical puede ser generalizada para listar cualquier nimero de elementos
seguidos por una “I” y la lista de elementos restantes.

?- [a,b,c,d] = [a,b|[c,d]].
yes

?- [a,b,c,d] = [a,blc,d].
no

?- [1,2,3,4,5] = [A,B,C|T].
A 1

B =2

cC =3

T = [4,5]

?— [al[b,c]] la,bl[c]].
yes

7.1. Operaciones sobre listas

Una operacién importante al trabajar con listas es aquella que determina si un objeto es un elemento
de la lista. Supongamos que para implementar la operacién de membresia utilizamos la relacion:

member (X, L)

donde X es un objeto y L es una lista. El objetivo member (X, L) es verdadero si X ocurre en L.
Ejemplo:

?— member (b, [a,b,c]) .
yes

?— member (b, [a, [b,c]]) .
no

?— member ([b,c], [a, [b,c]]).
yes

El programa Prolog que implementa la relacién member es:

member (X, [X|_]1) .
member (X, [_|1T]) :— member (X,T).

Ejercicio: Ejecute manualmente la consulta ?— member (R, [1,2]) .

Otra operacion util sobre listas es aquella que, dadas dos listas, obtiene una tercera lista que repre-
senta la concatenacién de las dos primeras. Ejemplo:



?— concat([a,b,c],[1,2,3],L).
L = [a,b,c,1,2,3]

?—- concat([a,b,c],I[1,2,3],[a,b,c,1,2,3]).
yes

El programa Prolog que implementa la operacién concat es:

concat ([],L2,L2).
concat ([X|L1l],L2, [X|L3]) :— concat(Ll,L2,L3).

Un aspecto interesante de esta relacion es que se puede hacer una consulta:

?— concat (L, [a,b], [1,2,a,b]).
L = 1[1,2]

Esto muestra un aspecto que dificilmente es logrado por otros lenguajes de programacién. Las rela-
ciones permiten obtener resultados, dados ciertos datos de entrada. Lo sorprendente es que en algunos
casos, dado un resultado, se puede determinar cuales serian los datos de entrada que permiten obtenerlo.



